AutoCAD VBA – Three-day course

Outline
This exciting and challenging "hands on" course focuses on building a “usable” user-interface, working with the AutoCAD object model, processing advanced-filtered-selection-sets, and VBA programming techniques

Course Objectives
By the end of the course the student will have mastered the skills required to create and edit AutoCAD objects.  The student will have a firm grasp of the AutoCAD object model and the methods and properties of working with AutoCAD drawing elements.  The student will also have a good understanding of “best practices” for coding a VBA application and be able to create a VBA program.

Course Content

Course Expectations…  What do you want from this course?

Excel databases quicker drawings aviodd retetiive stuff.  VB interfacing with equipment

Brain storming session on what the class would like to create for a simple VBA program.

Your ideas go here!!!

Lookng for a new way to do things.  Communication with outher aplications.  Customerns premanufactured houssing.  Panelaize the whole building.  Automation!!!!!  DATABASES.  Bill of materials.  Make cool stuff very very fast and kick but!  File management. Pulll info from one app to another…

Form

Module 

Proceedure

Terminoligy.

Interact with the user.

How to setup AutoCAD for VBA.

To use VBA for AutoCAD requires AutoCAD r14.01. If you do not have the 14.01 version see your local autodesk dealer for details on how to obtain the r14.01 update. When installing or updating to AutoCAD r14.01 also select the install VBA option (default on Typical and Custom installation settings).  For AutoCAD 2000/I, 2002 users, make sure you install VBA (do a FULL install)

To start the VBA editor:

To start the VBA editor for a new project inside of AutoCAD do the following: 

From the pull down menu select: Tools | Macro | Visual Basic Editor 

Or at the command prompt type: VBAide 
Or for AutoCAD 2000/I, 2002 use : ALT + F11

The VBA interface:

With Visual Basic, the programmer creates a user interface by adding controls from the Toolbox to a Form. To create a push button only requires a selection of the button control in the Toolbox and drawing/placing the control on a Form. It is very easy to create a Form that looks the way it will be used as a program. Buttons to push, Text boxes for entry, Option lists, etc, are a breeze to create, size, and change. The ability to rapidly create a prototype interface for a new program is stunning to say the least. Also, the resulting Form can be saved and imported into any Windows VBA-enabled program. 

Creating User Forms (talking to the user) 

Working with data from a Form

EVENTS

Nothing happens on a Form unless an Event occurs.

Take for instance, the Mouse Click event on a Form.

This EVENT occurs when the user presses and then releases a mouse button over an object. It can also occur when the value of a control is changed.

For a Form object, this event occurs when the user clicks either a blank area or a disabled control. For a control, this event occurs when the user: 
· Clicks a control with the left or right mouse button. With a CheckBox, CommandButton, Listbox, or OptionButton control, the Click event occurs only when the user clicks the left mouse button.

· Selects an item in a ComboBox or ListBox control, either by pressing the arrow keys or by clicking the mouse button.

· Presses the SPACEBAR when a CommandButton, OptionButton, or CheckBox control has the focus
.

· Presses ENTER when a form has a CommandButton control with its Default property set to True.

· Presses ESC when a form has a Cancel button — a CommandButton control with its Cancel property set to True.

· Presses an access key
 for a control. For example, if the caption of a CommandButton control is "&Go", pressing ALT+G triggers the event. 

You can also trigger the Click event in code by: 

Setting a CommandButton control's Value property to True.

Setting an OptionButton control's Value property to True.

Changing a CheckBox control's Value property setting. 

Click Event

Typically, you attach a Click event procedure to a CommandButton control, Menu object, or PictureBox control to carry out commands and command-like actions. For the other applicable controls, use this event to trigger actions in response to a change in the control.
You can use a control's Value property to test the state of the control from code. Clicking a control generates MouseDown and MouseUp events in addition to the Click event. The order in which these three events occur varies from control to control. For example, for ListBox and CommandButton controls, the events occur in this order: MouseDown, Click, MouseUp. But for FileListBox, Label, or PictureBox controls, the events occur in this order: MouseDown, MouseUp, and Click. When you're attaching event procedures for these related events, be sure that their actions don't conflict. If the order of events is important in your application, test the control to determine the event order.

Note   To distinguish between the left, right, and middle mouse buttons, use the MouseDown and MouseUp events.

If there is code in the Click event, the DblClick event will never trigger, because the Click event is the first event to trigger between the two. As a result, the mouse click is intercepted by the Click event, so the DblClick event doesn't occur.

Number crunching Form input

Okay, so you have a Form that the user wants to input data and run a calculation.

For instance, the user wants to calculate a right angle triangle using A²+B² = C².  To accommodate the user, you create a Form with Three Text boxes and one command button.

Form Code

Option Explicit

Private Sub CommandButton1_Click()

Dim dblA As Double

Dim dblB As Double

Dim dblC As Double

  dblA = TextBox1.Text

  dblA = dblA * dblA

  dblB = TextBox2.Text

  dblB = dblB * dblB

  dblC = Sqr(dblA + dblB)

  TextBox3.Text = dblC

End Sub

For a full listing of VBA Math Function see the AutoCAD Help menu.

Search for Math Functions (da!)
Declaring Variables & Variable types
Why?

When you declare that a variable is an object type, you setup the properties of the object.

Using Intellisense
, the properties of the object can then be accessed while you are writing your code.  Lets try it out.

This feature alone makes coding in VB/VBA very glamorous/easy!
Introduction to Variables, Constants and Data Types

You often need to store values temporarily when performing calculations with Visual Basic. For example, you might want to calculate several values, compare them, and perform different operations on them, depending on the result of the comparison. You need to retain the values if you want to compare them, but you don't need to store them in a property.
Visual Basic, like most programming languages, uses variables for storing values. Variables have a name (the word you use to refer to the value the variable contains) and a data type (which determines the kind of data the variable can store). Arrays can be used to store indexed collections of related variables.

Constants also store values, but as the name implies, those values remain constant throughout the execution of an application. Using constants can make your code more readable by providing meaningful names instead of numbers. There are a number of built-in constants in Visual Basic, but you can also create your own.

Data types control the internal storage of data in Visual Basic. By default, Visual Basic uses the Variant data type. There are a number of other available data types that allow you to optimize your code for speed and size when you don't need the flexibility that Variant provides.

Variables
In Visual Basic, you use variables to temporarily store values during the execution of an application. Variables have a name (the word you use to refer to the value the variable contains) and a data type (which determines the kind of data the variable can store).
You can think of a variable as a placeholder in memory for an unknown value. For example, imagine you are creating a program for a fruit stand to track the sales of apples. You don't know the price of an apple or the quantity sold until the sale actually occurs. You can use two variables to hold the unknown values — let's name them ApplePrice and ApplesSold. Each time the program is run, the user supplies the values for the two variables. To calculate the total sales and display it in a Textbox named txtSales, your code would look like this:

txtSales.txt = ApplePrice * ApplesSold

The expression returns a different total each time, depending on what values the user provides. The variables allow you to make a calculation without having to know in advance what the actual inputs are.

In this example, the data type of ApplePrice is Currency; the data type of ApplesSold is an integer. Variables can represent many other values as well: text values, dates, various numeric types, even objects.

Storing and Retrieving Data in Variables

You use assignment statements to perform calculations and assign the result to a variable:

ApplesSold = 10      ' The value 10 is passed to the 

                  ' variable.

ApplesSold = ApplesSold + 1      ' The variable is 

                              ' incremented.

Note that the equal sign in this example is an assignment operator, not an equality operator; the value (10) is being assigned to the variable (ApplesSold).

Declaring Variables

To declare a variable is to tell the program about it in advance. You declare a variable with the Dim statement, supplying a name for the variable:

Dim variablename [As type]

Variables declared with the Dim statement within a procedure exist only as long as the procedure is executing. When the procedure finishes, the value of the variable disappears. In addition, the value of a variable in a procedure is local to that procedure — that is, you can't access a variable in one procedure from another procedure. These characteristics allow you to use the same variable names in different procedures without worrying about conflicts or accidental changes.

A variable name: 

Must begin with a letter.

Can't contain an embedded period or embedded type-declaration character.

Must not exceed 255 characters.

Must be unique within the same scope, which is the range from which the variable can be referenced — a procedure, a form, and so on. 

The optional As type clause in the Dim statement allows you to define the data type or object type of the variable you are declaring. Data types define the type of information the variable stores. Some examples of data types include String, Integer, and Currency. Variables can also contain objects from Visual Basic or other applications. Examples of Visual Basic object types, or classes, include Object, Form1, and TextBox. 

Option Explicit – Why and Wherefore.

Explicit Declaration
To avoid the problem of misnaming variables, you can stipulate that Visual Basic always warn you whenever it encounters a name not declared explicitly as a variable.

To explicitly declare variables 

Place this statement in the Declarations section of a class, form, or standard module: 

Option Explicit
–or–

From the Tools menu, choose Options, click the Editor tab and check the Require Variable Declaration option. This automatically inserts the Option Explicit statement in any new modules, but not in modules already created; therefore, you must manually add Option Explicit to any existing modules within a project.

Had this statement been in effect for the form or standard module containing the SafeSqr function, Visual Basic would have recognized TempVal and TemVal as undeclared variables and generated errors for both of them. You could then explicitly declare TempVal:

Function SafeSqr(num)

   Dim TempVal

   TempVal = Abs(num)

   SafeSqr = Sqr(TemVal)

End Function

Now you'd understand the problem immediately because Visual Basic would display an error message for the incorrectly spelled TemVal. Because the Option Explicit statement helps you catch these kinds of errors, it's a good idea to use it with all your code.

Note   The Option Explicit statement operates on a per-module basis; it must be placed in the Declarations section of every form, standard, and class module for which you want Visual Basic to enforce explicit variable declarations. If you select Require Variable Declaration, Visual Basic inserts Option Explicit in all subsequent form, standard, and class modules, but does not add it to existing code. You must manually add Option Explicit to any existing modules within a project.

Option Explicit Statement Example
This example uses the Option Explicit statement to force explicit declaration of all variables. Attempting to use an undeclared variable causes an error at compile time. The Option Explicit statement is used at the module level only.

Option explicit   ' Force explicit variable declaration.

Dim MyVar   ' Declare variable.

MyInt = 10   ' Undeclared variable generates error.

MyVar = 10   ' Declared variable does not generate error.

Declaring AutoCAD objects in VBA (early binding) 

To declare an object to be a specific AutoCAD object use Dim to declare the object as an AutoCAD object.  Use the following format:

Dim [SomeObjectName] as [SomeAutoCADobjectType]

To declare an object as an AutoCAD layer use the following:

Dim objLayer as AcadLayer 

Example code to display the layer names in a drawing

Sub LayerNames()

    Dim objLayer As AcadLayer

    Dim objLayers As AcadLayers

    Dim strLayers As String

    Set objLayers = ThisDrawing.Layers

    For Each objLayer In objLayers

    strLayers = strLayers & objLayer.Name & vbCr

    Next

    MsgBox strLayers, vbInformation, "Layers in Drawing"

End Sub

Accessing the AutoCAD object model

Using VBA from inside of AutoCAD, the object model loaded and just waiting for you to declare (Dim objLayer as ACADLAYER) what your object is.

From inside AutoCAD’s VBAide

Tools | References…

Accessing the AutoCAD object model from VB 

Accessing the AutoCAD object model from VB requires the additional step of creating a reference to the AutoCAD object Model 

From VB

Project | References…

Describing Objects in VBA.
VBA uses Objects to work within documents. VBA describes everything within an AutoCAD drawing (lines, circles, arcs, text, etc) as an Object. Every Object in a drawing then has properties. A line within the current drawing in model space has a color property that would be described as Thisdrawing.ModelSpace.lineobject.color . A full listing and graphical representation of the AutoCAD Object model as well as Methods, Properties, Constants, VBA events and example code — can be accessed in AutoCAD from the online help (press F1 help) as shown below.

Creating new AutoCAD drawing objects from VBA 

To create a new object in an AutoCAD drawing from VBA requires the programmer to know what “space” (model or paper) to create the object.  The object is created using the add method.

Example code to add a circle in an AutoCAD drawing

Sub CreateCircle()

    Dim Center As Variant

    Dim Radiaus As Double

    Dim objCircle As AcadCircle

    Radius = 5

    ' prompt user for circle location

    Center = ThisDrawing.Utility.GetPoint(, "Pick location for Centre of Circle:")

    Set objCircle = ThisDrawing.ModelSpace.AddCircle(Center, Radius)

    objCircle.Color = acBlue

    objCircle.Update

End Sub

Modifying AutoCAD objects

Sample programs:

Sub REFdim()

‘ Select dimensions and add  REF. To the dimensions

    Dim objDim As AcadDimension

    Dim newss As AcadSelectionSet

    Dim FilterType(0) As Integer

    Dim FilterData(0) As Variant

    FilterType(0) = 0

    FilterData(0) = "DIMENSION"

    Set newss = ThisDrawing.SelectionSets.Add("RefDims")

    newss.SelectOnScreen FilterType, FilterData

    For Each objDim In newss

    objDim.TextOverride = "<> REF."

    objDim.Update

    Next

    newss.Delete

End Sub

Sub TYPDim()

‘ This program selects only dimensions and adds “ TYP.”

    Dim objDim As AcadDimension

    Dim newss As AcadSelectionSet

    Dim FilterType(0) As Integer

    Dim FilterData(0) As Variant

    FilterType(0) = 0

    FilterData(0) = "DIMENSION"

    Set newss = ThisDrawing.SelectionSets.Add("RefDims")

    newss.SelectOnScreen FilterType, FilterData

    For Each objDim In newss

    objDim.TextOverride = "<> TYP."

    objDim.Update

    Next

    newss.Delete

End Sub

Creating a selection set (with/without selection set filters) 

Creating a simple selection set (without filters)

Example code

Sub SelectionSet()

' Create a selection set from objects selected on screen

' Display the OjbectName of each object in the selection set.

    Dim newSS As AcadSelectionSet

    Dim objEntitie As AcadObject

    Set newSS = ThisDrawing.SelectionSets.Add("MySelectionSet")

    newSS.SelectOnScreen

    ' Work the selection set

    For Each objEntitie In newSS

    
MsgBox objEntitie.ObjectName

    Next

    ' remove/free-up the selection set

    newSS.Delete

End Sub

A selection set with simple filters

Example code

Sub REFdim()

‘ Select dimensions and add  REF. To the dimensions

    Dim objDim As AcadDimension

    Dim newss As AcadSelectionSet

    Dim FilterType(0) As Integer

    Dim FilterData(0) As Variant

    FilterType(0) = 0

    FilterData(0) = "DIMENSION"
    Set newss = ThisDrawing.SelectionSets.Add("RefDims")

    newss.SelectOnScreen FilterType, FilterData
    For Each objDim In newss

    objDim.TextOverride = "<> REF."

    objDim.Update

    Next

    newss.Delete

End Sub

Exampe Code

Sub TYPDim()

‘ This program selects only dimensions and adds “ TYP.”

    Dim objDim As AcadDimension

    Dim newss As AcadSelectionSet

    Dim FilterType(0) As Integer

    Dim FilterData(0) As Variant

    FilterType(0) = 0

    FilterData(0) = "DIMENSION"
    Set newss = ThisDrawing.SelectionSets.Add("RefDims")

    newss.SelectOnScreen FilterType, FilterData
    For Each objDim In newss

    objDim.TextOverride = "<> TYP."

    objDim.Update

    Next

    newss.Delete

End Sub

Working a selection set (processing selected objects)

Creating A selection set with multiple filters

Example code

Sub rotate180()

    ' this program will rotate objects 180 degrees about the insert point

    ' This program will work with text and blocks.

    Dim newss As AcadSelectionSet

    Dim FilterType(0 To 4) As Integer

    Dim FilterData(0 To 4) As Variant

    Dim objEntity As AcadObject

    Dim dblRotAngle As Double

    FilterType(0) = -4

    FilterData(0) = "<OR"

    FilterType(1) = 0

    FilterData(1) = "TEXT"

    FilterType(2) = 0

    FilterData(2) = "MTEXT"

    FilterType(3) = 0

    FilterData(3) = "INSERT"

    FilterType(4) = -4

    FilterData(4) = "OR>"
   Set newss = vbdPowerSet("MySharonaAA")

    newss.SelectOnScreen FilterType, FilterData
    For Each objEntity In newss

        objEntity.rotate objEntity.InsertionPoint, AngleToRadians(180)

    Next

    newss.Delete

End Sub

Purging a selection set

VdbPowerSet
In AutoCAD 2000/i/2002, the name for a selection set must be unique or an error will be generated when you attempt to add it to the selection set collection. Use this to add a new selection set by name, and never see the existing selection set error again!

Courtesy of Randal Rath www.vbdesign.net 

Code:

'//VBD Power Set// 

Public Function vbdPowerSet(strName As String) As AcadSelectionSet

  Dim objSelSet As AcadSelectionSet

  Dim objSelCol As AcadSelectionSets

  Set objSelCol = ThisDrawing.SelectionSets

    For Each objSelSet In objSelCol

      If objSelSet.Name = strName Then

        objSelCol.Item(strName).Delete

        Exit For

      End If

    Next

  Set objSelSet = objSelCol.Add(strName)

  Set vbdPowerSet = objSelSet

End Function
'From The Llama Library Published November 12, 1999
Running a VBA macro from an AutoCAD Menu or Tool bar 

Integrating VBA programs with the AutoCAD Menu 

^C^C-VBARUN;C:/DD/CADSTAND.DVB!START.REFDIM;

To launch a VBA subprogram From AutoLisp

(COMMAND "-VBARUN" "C:/DD/CADSTAND.DVB!START.REFDIM")



Exploring “hidden” Object properties.

Not all object properties are exposed in the Intellisense properties list box.

For instance, in AutoCAD 2000, the Measurement property of a dimension is not listed in the propeties of the dimension.  To explose the properties of an object, in AutoCAD, use the properties dialouge box and select the object (in this case a dimension).

Using Breaks and Watches
Trouble shooting your code
Where to find FREE VBA reference material (code)


On The Internet

www.contractcaddgroup.com
www.vbdesign.net
www.vbcad.com
http://codeguru.earthweb.com/vb/index.shtml
http://www.vb-helper.com/index.htm
http://www.vbapro.com/
http://www.planet-source-code.com/vb/

Time permitting…

Building the DAO connection in VBA to the database

Loading the Microsoft DAO 3.6 project reference

To hook into an Access Data base from VBA requires loading the reference to the Microsoft DAO Object Library.

Robert Cheek’s Description of DAO goes here.

Lines below intentionally left blank for YOU to fill in with your own notes.

Example code to suck-data from an Access database.

Declaring the database object and database location in code.

Sub StudentDatabase()

    Dim Mydatabase As Database

    Dim SQLtext As String

    Dim qdfTemp As QueryDef

    Dim rstTemp As Recordset

    Dim iRecordCount As Integer

    Dim I As Integer

    Dim strMtext As String

    Dim Pt1(0 To 2) As Double

    SQLtext = "SELECT Name, Mark, Phone" & _

                " FROM Student" & _

                " ORDER BY Mark DESC;"

    ' Set the Database via a path

    Set Mydatabase = OpenDatabase("c:\DD\vba.mdb")

    Set qdfTemp = Mydatabase.CreateQueryDef("", SQLtext)

    Set rstTemp = qdfTemp.OpenRecordset(dbOpenDynaset)

    If rstTemp.RecordCount = 0 Then GoTo NeetExit

    ' Goto the last record in the record set

    rstTemp.MoveLast

    ' How many Records in the record set

    iRecordCount = rstTemp.RecordCount

    ' Move to the First Record

    rstTemp.MoveFirst

    For I = 1 To iRecordCount

        strMtext = strMtext & "\P" & _

rstTemp!Name & " " & _

rstTemp!Mark & " " & _

rstTemp!phone

        'MsgBox rstTemp!Name & " " & rstTemp!Mark & " " & rstTemp!phone, vbInformation, "Name Mark and Phone Number from Database"

        rstTemp.MoveNext

    Next

    ' Debug.Print strMtext

    Pt1(0) = 0 ' X

    Pt1(1) = 0 ' Y

    Pt1(2) = 0 ' Z

      ThisDrawing.ModelSpace.AddMText Pt1, 5, strMtext

      ThisDrawing.Application.ZoomExtents

NeetExit:

     Mydatabase.Close

End Sub

Creating a SQL query to pull information from a database

Short cut tips on building a SQL query

To build a Structured Query Language (SQL) Query (question) from scratch can be an arduous task at best.  One of the easier ways to create an SQL query is to start from within Access.  

1. From within Access open a table (an existing Access table or link a table from an other database – Oracle, Excel, Sequel Server, etc).

2. Open a new Query and select the table to work with

3. Next, create your Query and test it out…

4. After you have your query generating the  data you are searching for, take a look at the SQL that Access is using.  To see the SQL information (used by the query) use the pulldown menu 
View | SQL
5. Next comes the tricky bit...  Getting the SQL statement to work in VBA/VB.  As you have seen through out the process of storing information in VB/VBA, the information that we need to store for a SQL query is a TEXT string.  However, to complicate matters, we have a text string that also makes use of " – Oh Oh!!  The " is typically used at the beginning and END of a text string in VBA/VB.  So what we need to do is store a text string to a variable and also include " within the text string.  Below are two examples of how this task could be accomplished.

Dim strSQLtext As String

Dim Qt As String

Qt = Chr(34) ' "

strSQLtext = "SELECT Name, Mark, Phone" & _

              " FROM Student" & _

              " WHERE (((Name) Like ""*Frank*""));"

strSQLtext = "SELECT Name, Mark, Phone" & _

              " FROM Student" & _

              " WHERE (((Name) Like " & Qt & "*Frank*" & Qt & "));"

Note: if you want to return ALL the fields of a record use SELECT *

strSQLtext = "SELECT *" & _

              " FROM Student" & _

              " WHERE (((Name) Like ""*Frank*""));"
Editing/creating drawings that are not open using AutoCAD/ObjectDBX 

“Ever wished you could look into a drawing without opening it? Now you can. You need AutoCAD 2000 to make this work (the dll shipped with it) or a separate license from AutoDesk.” -- Randal Rath www.vbdesign.net
My initial question with regards to ObjectDBX was why bother?  If you can automate the processing drawing in AutoCAD, why use ObjectDBX?  The answer turns out to be  SPEED.  ObjectDBX does not load the drawing into the viewable portion of AutoCAD and therefore things like regens and graphics updates are not required – very cool! 
Setting up ObjectDBX

Using AxDb15.dll

1) Start a new VBA project

2) Add a form - any name

3) Add a reference to the dll

3A) Go to the "Tools" menu

3B) Go to "References" <click>

3C) In the dialog look for "object DBX 1.0 Type Library" <check>

3D) OK button <click>

Create the form above using:

1 Text box

1 combo box

1 list box

1 command button

Then add the following code:

'Begin code Block

Option Explicit

Dim objDbx As AxDbDocument

Private Sub ComboBox1_Change()

Dim objGeneral As Object

ListBox1.Clear

Select Case ComboBox1.Text

Case "Layers"

For Each objGeneral In objDbx.Layers

ListBox1.AddItem objGeneral.Name

Next

Case "Blocks"

For Each objGeneral In objDbx.Blocks

ListBox1.AddItem objGeneral.Name

Next

Case "X - Refs"

For Each objGeneral In objDbx.Blocks

 If objGeneral.IsXRef = True Then

  ListBox1.AddItem objGeneral.Name

 End If

Next

End Select

End Sub

Private Sub TextBox1_Change()

CommandButton1.Enabled = True

End Sub

Private Sub UserForm_Initialize()

Set objDbx = GetInterfaceObject("ObjectDBX.AxDbDocument")

CommandButton1.Caption = "Load Drawing Info"

ComboBox1.Style = fmStyleDropDownList

ComboBox1.AddItem "Layers"

ComboBox1.AddItem "Blocks"

ComboBox1.AddItem "X - Refs"

ComboBox1.Enabled = False

Me.Caption = "Working with AxDb15"

End Sub

Private Sub CommandButton1_Click()

Dim strPath As String

strPath = TextBox1.Text

  If Dir(strPath) <> "" Then

    objDbx.Open strPath

    ComboBox1.Enabled = True

    CommandButton1.Enabled = False

  Else

    MsgBox "The drawing " & strPath & " is not valid", vbCritical, "Hummm..."

  End If

End Sub

'End Code Block

Porting a VBA program to a standalone VB EXE
To update the coding examples for use with VB, you must first reference the AutoCAD type library. To do this in VB, select the References option from the Project menu to launch the Reference dialog box. From the References dialog box, choose AutoCAD Type Library, and then press OK. 
Next, in the code example replace all references to ThisDrawing with a user-specified variable referencing the active document. To do this, define a variable for the AutoCAD application (acadApp) and for the current document (acadDoc). Then, set the application variable to the current AutoCAD application. 

If AutoCAD is running, the VB GetObject function retrieves the AutoCAD Application object. If AutoCAD is not running, an error occurs that (in this example) is trapped, then cleared. The CreateObject function then attempts to create an AutoCAD Application object. If it succeeds, AutoCAD is started; if it fails, a message box displays a description of the error. 

Note: When running multiple sessions of AutoCAD, the GetObject function will return the first instance of AutoCAD in the Windows Running Object Table. See the Microsoft Visual Basic documentation on the Running Object Table (ROT) and the GetObject function for more information on verifying the session returned by GetObject. 

You must set the AutoCAD application's Visible property to TRUE in order to display the AutoCAD drawing window. 

Note: If GetObject creates a new instance of AutoCAD (that is, AutoCAD was not already running when you issued GetObject), failure to set Visible to TRUE results in an invisible AutoCAD application; AutoCAD will not even appear on the Windows taskbar. 

The following code example demonstrates creating a line in both VBA and VB. 
Creating a line using VBA: 

Sub AddLineVBA()

    ' This example adds a line

    ' in model space

    Dim lineObj As AcadLine

    Dim startPoint(0 To 2) As Double

    Dim endPoint(0 To 2) As Double

    ' Define the start and end

    ' points for the line

    startPoint(0) = 1

    startPoint(1) = 1

    startPoint(2) = 0

    endPoint(0) = 5

    endPoint(1) = 5

    endPoint(2) = 0

    ' Create the line in model space

    Set lineObj = ThisDrawing. _

        ModelSpace.AddLine _

        (startPoint, endPoint)

    ' Zoom in on the newly created line

    ZoomAll

End Sub

Creating a line using VB: 

Sub AddLineVB()

    On Error Resume Next

    ' Connect to the AutoCAD application

    Dim acadApp As AcadApplication

    Set acadApp = GetObject _

                  (, "AutoCAD.Application")

    If Err Then

        Err.Clear

        Set acadApp = CreateObject _

                  ("AutoCAD.Application")

        If Err Then

            MsgBox Err.Description

            Exit Sub

        End If

    End If

    ' Connect to the AutoCAD drawing

    Dim acadDoc As AcadDocument

    Set acadDoc = acadApp.ActiveDocument

    ' Establish the endpoints of the line

    Dim lineObj As AcadLine

    Dim startPoint(0 To 2) As Double

    Dim endPoint(0 To 2) As Double

    startPoint(0) = 1

    startPoint(1) = 1

    startPoint(2) = 0

    endPoint(0) = 5

    endPoint(1) = 5

    endPoint(2) = 0

    ' Create a Line object in model space

    Set lineObj = acadDoc.ModelSpace.AddLine _

                         (startPoint, endPoint)

    ZoomAll

    acadApp.visible = True

End Sub

Apendex

Math Functions
ABS Function Example
This example uses the Abs function to compute the absolute value of a number.

Dim MyNumber

MyNumber = Abs(50.3)   ' Returns 50.3.

MyNumber = Abs(-50.3)   ' Returns 50.3.

Atn Function Example
This example uses the Atn function to calculate the value of pi.

Dim pi

pi = 4 * Atn(1)   ' Calculate the value of pi.

Cos Function Example
This example uses the Cos function to return the cosine of an angle.

Dim MyAngle, MySecant

MyAngle = 1.3   ' Define angle in radians.

MySecant = 1 / Cos(MyAngle)   ' Calculate secant.

Exp Function Example
This example uses the Exp function to return e raised to a power.

Dim MyAngle, MyHSin

' Define angle in radians.

MyAngle = 1.3   

' Calculate hyperbolic sine.

MyHSin = (Exp(MyAngle) - Exp(-1 * MyAngle)) / 2   
Int Function, Fix Function Example
This example illustrates how the Int and Fix functions return integer portions of numbers. In the case of a negative number argument, the Int function returns the first negative integer less than or equal to the number; the Fix function returns the first negative integer greater than or equal to the number.

Dim MyNumber

MyNumber = Int(99.8)   ' Returns 99.

MyNumber = Fix(99.2)   ' Returns 99.

MyNumber = Int(-99.8)   ' Returns -100.

MyNumber = Fix(-99.8)   ' Returns -99.

MyNumber = Int(-99.2)   ' Returns -100.

MyNumber = Fix(-99.2)   ' Returns -99.

Log Function Example
This example uses the Log function to return the natural logarithm of a number.

Dim MyAngle, MyLog

' Define angle in radians.

MyAngle = 1.3

' Calculate inverse hyperbolic sine.

MyLog = Log(MyAngle + Sqr(MyAngle * MyAngle + 1))
Rnd Function Example
This example uses the Rnd function to generate a random integer value from 1 to 6.

Dim MyValue

MyValue = Int((6 * Rnd) + 1)   ' Generate random value between 1 and 6.

Sgn Function Example
This example uses the Sgn function to determine the sign of a number.

Dim MyVar1, MyVar2, MyVar3, MySign

MyVar1 = 12: MyVar2 = -2.4: MyVar3 = 0

MySign = Sgn(MyVar1)   ' Returns 1.

MySign = Sgn(MyVar2)   ' Returns -1.

MySign = Sgn(MyVar3)   ' Returns 0.

Sin Function Example
This example uses the Sin function to return the sine of an angle.

Dim MyAngle, MyCosecant

MyAngle = 1.3   ' Define angle in radians.

MyCosecant = 1 / Sin(MyAngle)   ' Calculate cosecant.

Sqr Function Example
This example uses the Sqr function to calculate the square root of a number.

Dim MySqr

MySqr = Sqr(4)   ' Returns 2.

MySqr = Sqr(23)   ' Returns 4.79583152331272.

MySqr = Sqr(0)   ' Returns 0.

MySqr = Sqr(-4)   ' Generates a run-time error.

Tan Function Example
This example uses the Tan function to return the tangent of an angle.

Dim MyAngle, MyCotangent

MyAngle = 1.3   ' Define angle in radians.

MyCotangent = 1 / Tan(MyAngle)   ' Calculate cotangent.

Logical Operators

And Operator Example
This example uses the And operator to perform a logical conjunction on two expressions.

Dim A, B, C, D, MyCheck

A = 10: B = 8: C = 6: D = Null   ' Initialize variables.

MyCheck = A > B And B > C   ' Returns True.

MyCheck = B > A And B > C   ' Returns False.

MyCheck = A > B And B > D   ' Returns Null.

MyCheck = A And B   ' Returns 8 (bitwise comparison).

Eqv Operator Example
This example uses the Eqv operator to perform logical equivalence on two expressions.

Dim A, B, C, D, MyCheck

A = 10: B = 8: C = 6: D = Null   ' Initialize variables.

MyCheck = A > B Eqv B > C   ' Returns True.

MyCheck = B > A Eqv B > C   ' Returns False.

MyCheck = A > B Eqv B > D   ' Returns Null.

MyCheck = A Eqv B   ' Returns -3 (bitwise comparison).

Imp Operator Example
This example uses the Imp operator to perform logical implication on two expressions.

Dim A, B, C, D, MyCheck

A = 10: B = 8: C = 6: D = Null   ' Initialize variables.

MyCheck = A > B Imp B > C   ' Returns True.

MyCheck = A > B Imp C > B   ' Returns False.

MyCheck = B > A Imp C > B   ' Returns True.

MyCheck = B > A Imp C > D   ' Returns True.

MyCheck = C > D Imp B > A   ' Returns Null.

MyCheck = B Imp A   ' Returns -1 (bitwise comparison).

Not Operator Example
This example uses the Not operator to perform logical negation on an expression.

Dim A, B, C, D, MyCheck

A = 10: B = 8: C = 6: D = Null   ' Initialize variables.

MyCheck = Not(A > B)   ' Returns False.

MyCheck = Not(B > A)   ' Returns True.

MyCheck = Not(C > D)   ' Returns Null.

MyCheck = Not A   ' Returns -11 (bitwise comparison).

Or Operator Example
This example uses the Or operator to perform logical disjunction on two expressions.

Dim A, B, C, D, MyCheck

A = 10: B = 8: C = 6: D = Null   ' Initialize variables.

MyCheck = A > B Or B > C   ' Returns True.

MyCheck = B > A Or B > C   ' Returns True.

MyCheck = A > B Or B > D   ' Returns True.

MyCheck = B > D Or B > A   ' Returns Null.

MyCheck = A Or B   ' Returns 10 (bitwise comparison).

Xor Operator Example
This example uses the Xor operator to perform logical exclusion on two expressions.

Dim A, B, C, D, MyCheck

A = 10: B = 8: C = 6: D = Null   ' Initialize variables.

MyCheck = A > B Xor B > C   ' Returns False.

MyCheck = B > A Xor B > C   ' Returns True.

MyCheck = B > A Xor C > B   ' Returns False.

MyCheck = B > D Xor A > B   ' Returns Null.

MyCheck = A Xor B   ' Returns 2 (bitwise comparison).

Decision Structures
Visual Basic procedures can test conditions and then, depending on the results of that test, perform different operations. The decision structures that Visual Basic supports include: 

· If...Then

· If...Then...Else

· Select Case 

If...Then

Use an If...Then structure to execute one or more statements conditionally. You can use either a single-line syntax or a multiple-line block syntax:

If condition Then statement
If condition Then
statements
End If
The condition is usually a comparison, but it can be any expression that evaluates to a numeric value. Visual Basic interprets this value as True or False; a zero numeric value is False, and any nonzero numeric value is considered True. If condition is True, Visual Basic executes all the statements following the Then keyword. You can use either single-line or multiple-line syntax to execute just one statement conditionally (these two examples are equivalent):

If anyDate < Now Then anyDate = Now

If anyDate < Now Then

   anyDate = Now

End If

Notice that the single-line form of If...Then does not use an End If statement. If you want to execute more than one line of code when condition is True, you must use the multiple-line block If...Then...End If syntax.

If anyDate < Now Then

   anyDate = Now

   Timer1.Enabled = False      ' Disable timer control.

End If

If...Then...Else

Use an If...Then...Else block to define several blocks of statements, one of which will execute:

If condition1 Then
[statementblock-1]
[ElseIf condition2 Then
[statementblock-2]] ...
[Else
[statementblock-n]]

End If
Visual Basic first tests condition1. If it's False, Visual Basic proceeds to test condition2, and so on, until it finds a True condition. When it finds a True condition, Visual Basic executes the corresponding statement block and then executes the code following the End If. As an option, you can include an Else statement block, which Visual Basic executes if none of the conditions are True.

If...Then…ElseIf is really just a special case of If...Then...Else. Notice that you can have any number of ElseIf clauses, or none at all. You can include an Else clause regardless of whether you have ElseIf clauses.

For example, your application could perform different actions depending on which control in a menu control array was clicked:

Private Sub mnuCut_Click (Index As Integer)

   If Index = 0 Then            ' Cut command.

      CopyActiveControl         ' Call general procedures.

      ClearActiveControl

   ElseIf Index = 1 Then      ' Copy command.

      CopyActiveControl

   ElseIf Index = 2 Then      ' Clear command.

      ClearActiveControl

   Else                        ' Paste command.

      PasteActiveControl

   End If

End Sub

Notice that you can always add more ElseIf parts to your If...Then structure. However, this syntax can get tedious to write when each ElseIf compares the same expression to a different value. For this situation, you can use a Select Case decision structure.

Select Case

Visual Basic provides the Select Case structure as an alternative to If...Then...Else for selectively executing one block of statements from among multiple blocks of statements. A Select Case statement provides capability similar to the If...Then...Else statement, but it makes code more readable when there are several choices.

A Select Case structure works with a single test expression that is evaluated once, at the top of the structure. Visual Basic then compares the result of this expression with the values for each Case in the structure. If there is a match, it executes the block of statements associated with that Case:

Select Case testexpression
[Case expressionlist1
[statementblock-1]]
[Case expressionlist2
[statementblock-2]]
.
.
.
[Case Else
[statementblock-n]]

End Select
Each expressionlist is a list of one or more values. If there is more than one value in a single list, the values are separated by commas. Each statementblock contains zero or more statements. If more than one Case matches the test expression, only the statement block associated with the first matching Case will execute. Visual Basic executes statements in the Case Else clause (which is optional) if none of the values in the expression lists matches the test expression.

For example, suppose you added another command to the Edit menu in the If...Then...Else example. You could add another ElseIf clause, or you could write the function with Select Case:

Private Sub mnuCut_Click (Index As Integer)

   Select Case Index

      Case 0                  ' Cut command.

         CopyActiveControl      ' Call general procedures.

         ClearActiveControl

      Case 1                  ' Copy command.

         CopyActiveControl

      Case 2                  ' Clear command.

         ClearActiveControl

      Case 3                  ' Paste command.

         PasteActiveControl

      Case Else

         frmFind.Show         ' Show Find dialog box.

   End Select

End Sub

Notice that the Select Case structure evaluates an expression once at the top of the structure. In contrast, the If...Then...Else structure can evaluate a different expression for each ElseIf statement. You can replace an If...Then...Else structure with a Select Case structure only if the If statement and each ElseIf statement evaluates the same expression.

Select Case Statement Example
This example uses the Select Case statement to evaluate the value of a variable. The second Case clause contains the value of the variable being evaluated, and therefore only the statement associated with it is executed.

Dim Number

Number = 8   ' Initialize variable.

Select Case Number   ' Evaluate Number.

Case 1 To 5   ' Number between 1 and 5, inclusive.

   Debug.Print "Between 1 and 5"

' The following is the only Case clause that evaluates to True.

Case 6, 7, 8   ' Number between 6 and 8.

   Debug.Print "Between 6 and 8"

Case 9 To 10   ' Number is 9 or 10.

Debug.Print "Greater than 8"

Case Else   ' Other values.

   Debug.Print "Not between 1 and 10"

End Select
While...Wend Statement Example
This example uses the While...Wend statement to increment a counter variable. The statements in the loop are executed as long as the condition evaluates to True.

Dim Counter

Counter = 0   ' Initialize variable.

While Counter < 20   ' Test value of Counter.

   Counter = Counter + 1   ' Increment Counter.

Wend   ' End While loop when Counter > 19.

Debug.Print Counter   ' Prints 20 in the Immediate window.

For...Next Statement Example
This example uses the For...Next statement to create a string that contains 10 instances of the numbers 0 through 9, each string separated from the other by a single space. The outer loop uses a loop counter variable that is decremented each time through the loop.

Dim Words, Chars, MyString

For Words = 10 To 1 Step -1    ' Set up 10 repetitions.

   For Chars = 0 To 9   ' Set up 10 repetitions.

      MyString = MyString & Chars   ' Append number to string.

   Next Chars   ' Increment counter

   MyString = MyString & " "   ' Append a space.

Next Words
For Each...Next Statement Example
This example uses the For Each...Next statement to search the Text property of all elements in a collection for the existence of the string "Hello". In the example, MyObject is a text-related object and is an element of the collection MyCollection. Both are generic names used for illustration purposes only.

Dim Found, MyObject, MyCollection

Found = False   ' Initialize variable.

For Each MyObject In MyCollection   ' Iterate through each element. 

   If MyObject.Text = "Hello" Then   ' If Text equals "Hello".

      Found = True   ' Set Found to True.

      Exit For   ' Exit loop.

   End If

Next

� Focus


The ability to receive mouse clicks or keyboard input at any one time. In the Microsoft Windows environment, only one window, form, or control can have this ability at a time. The object that "has the focus" is normally indicated by a highlighted caption or title bar. The focus can be set by the user or by the application.





� Access key


A key pressed while holding down the ALT key that allows the user to open a menu, carry out a command, select an object, or move to an object. For example, alt+F opens the File menu.





� IntelliSense


IntelliSense technology has now been brought to the developer, making it easier and faster than ever to write syntactically correct code in Visual Basic 5. IntelliSense features provide developers with instant syntax reference and object model assistance to reduce programming time and assure syntactically correct code.

